[bookmark: _Toc249321571][bookmark: _Toc249352772][bookmark: _Toc262039540][bookmark: _Toc264471433][bookmark: _Toc264516200][bookmark: _Toc264787825]Excel VBAによる表操作プログラム
TABLAS.xlsm
[bookmark: _Toc264516204][bookmark: _Toc264787829]最初の画面
　＊ショートカット[Ctrl]+[Shift]+TでTABLASのマクロを起動してください。次は「説明」のタブを開いたところです。「クリック：参照WEB」ボタンを押すと、ブラウザーが立ち上がり参照サイトを見ることができます。この画面では「実行」、「削除」のボタンは無効になっています。
 TABLAS.xlsを立ち上げると他のBook内のデータも分析できます。

「説明」タブ
　このフォームのタイトルバー（最上部の青くなっている部分）に、プログラム名(TABLAS. v.******)、最小化ボタン（―）、最大化ボタン（□）、終了ボタン（×）があります。分析中にこのフォームが邪魔になったときは最小化ボタンを押してください。そして最大化ボタンを押すと元の大きさに戻ります。終了ボタンを押すと、フォームが消えます。再度立ち上げるときはショートカット[Ctrl]+[Shift]+Tで起動してください。
　次の段には作者のメールアドレスが載せられています。プログラムの不具合や改善点などのご意見をお寄せください。なるべく多くの人に回答を差し上げられるようにいたします。
　「出力フォント」と「サイズ」のコンボボックスは、出力シートで選択されるフォント名とフォントサイズを示します。適当な項目を選択してください。
　「最大列幅」は出力データの列の最大幅を設定します。デフォルトは60ですが、これを大きくしたり、小さくしたりすることができます。論文やレポートなどに出力の画像をそのまま載せるときに、都合のよい大きさに設定してください。
　右下に「タイム」があり、TABLASを実行すると処理時間がここに表示されます。
　次はTABLASのエクセルシートに載せたサンプルデータです。

　下に、たとえば「E.sep」のようなシート名があります。プログラムの説明をするとき、このシート名でデータ例を示します。旧約聖書『創世記』各国語版の最初の3節、および5世紀にヒエロニムスがラテン語に訳し、16世紀にスペインで出版された版をデジタル化したVulgataです。

テキストをエクセルにインポート

インポート
　「◎実行」ボタンによって処理を開始します。出力したシートが不要であるときは「▲削除」ボタンで削除できます[footnoteRef:1]。右下のプログレスバーで実行の進行状況が示され、実行が終了したときに実行時間が表示されます。 [1:  エクセルのシートのタブを右クリックして削除することもできますが、これには確認のメッセージが現れます。一方TABLASの「削除」ボタンは確認をせずそのままシートを削除し、フォームの内容が更新されます。立ち上げたときに使われていたシートはTABLASからは削除できません。] 

段落で改行
　これから行う「資料の整理」を行うために、はじめに言語資料（コーパス）を用意しなければなりません。自分が持っている言語データや自分が作成した言語データをエクセルシート上にコピーして言語資料を作成しましょう。たとえば次のような形式のファイルをエクセルに読み込むことができます。
1.1  Madrid
	A la recepción de un hotel madrileño llega un profesor extranjero para participar como conferenciante en un seminario sobre Nutrición organizado por una universidad de verano con sede en El Escorial. El profesor hablará con el conserje, pidiéndole información sobre los servicios del hotel, así como sobre posibles visitas turísticas por la región.
　はじめに次の手順で「コピー」をしてください。コピー・アンド・ペーストという作業をするときは、はじめにコピーする対象を選択し[footnoteRef:2]、リボンの「ホーム」→「コピー」をクリックします。この瞬間に選択された内容が「クリップボード」に入ります[footnoteRef:3]。パソコンではペーストするまでは何も見えません。しかし、コピーした内容はちゃんとクリップボードに入っています。次に「実行ボタン」を押すと次のような結果になります[footnoteRef:4]。 [2:  対象のファイル名をマウスでドラッグしたり、[Shift]キーを押しながら→や↓などのカーソル移動キーを押すと選択状態になります。全部選択するときは[Ctrl]+Aを押します。]  [3:  ふつうクリップボードは見えませんが、それを「貼り付け」（ペースト）することによって内容がわかります。「コピー」のショートカットキーは[Ctrl]+C, 「ペースト」は[Ctrl]+Vです。このように頻繁に行う動作はなるべくショートカットキーを使うことを勧めます。]  [4:  ここでは、データ例として今執筆中のこの原稿をコピーしました。] 

文末で改行
「文末で改行します」のオプションを選択すると、和文や欧文の文末で改行します。次のように、改行して区切られたデータが別々のセルに配置されます。

[bookmark: _Toc264471439][bookmark: _Toc264516210][bookmark: _Toc264787835]段落と文末で改行
「段落と文末で改行します」のオプションを選択すると、和文や欧文の文末で改行します。段落番号と文番号を出力します。

カラムを指定
次のような形式（#…(見出し番号):…(見出し)）のファイルをエクセルに読み込むことができます。
	#1:0001
#2:1251
#3:Sevilla
#4:JPB
(1) Coñocida cosa sea a todos los que esta carta vieren cómo yo don Ferrando, por la gracia de Dios rey de Castiella, de Toledo, de León, de Gallizia, de Sevilla, de Córdova, de Murcia e de Jaén, embié (2) mis cartas a vós el concejo de Guadalfajara que embiássedes vuestros omnes buenos de vuestro concejo a mí, por cosas que avía de veer e de fablar convusco por buen paramiento de vuestra villa. E vós embiastes vuestros (3) omnes buenos ante mí, e yo fablé con ellos aquellas cosas que entendí que eran buen paramiento de la tierra. (…)

#1:0002
#2:1260
#3:Córdova
#4:X
(1) Sepan todos los ombres que esta carta vieren e oyeren cómo nós don Alfonso, por la gracia de Dios rey de Castiella, de Toledo, de León, de Gallicia, de Sevilla, de Córdova, de Murcia, (2) de Jaén e del Algarve, en uno con la reína doña Yolant mi mugier, e con nuestro fijo el infante don Fernando primero heredero, e con nuestros fijos el infante don Sancho e el infan(3)te don Pedro, por sabor que avemos de fazer bien e merced a todos los vezinos e moradores de la villa de Guadalfajara por muchos servicios que fizieron a nós e a nuestro linage, dámosles e (…)
otorgámosles que fagan dos ferias en la villa sobredicha de Guadalfajara por siempre jamás, e que las fagan dos vezes en el año, la una feria por Cinquaesma e que dure depués de la Cinquaesma quinze días, e (5) la otra feria que sea por sant Lucas, e comience ocho días ante de sant Lucas e dure ocho días después. (…)


(1)「#数字:」はテキストの属性を示します。#1は第1カラムにコロン(:)以下にあるデータが入ることを示します。#2以下も同様です。「#数字:」がない行をテキストとします。次が実行結果です。

(2) 階層化されたテキストをインポートするときは、次のように階層の内容が変わる場所だけを指定します。
[bookmark: _Toc341936869][bookmark: OLE_LINK1]#1:1.Hotel
1 En el hotel
#2:1.Mad
1.1 Madrid
	A la recepción de un hotel madrileño llega un profesor extranjero para participar como conferenciante en un seminario sobre Nutrición organizado por una universidad de verano con sede en El Escorial. El profesor hablará con el conserje, pidiéndole información sobre los servicios del hotel, así como sobre posibles visitas turísticas por la región. (...)
[bookmark: _Toc341936870]#2:2.Sev
1. 2 Sevilla
	En la recepción de un hotel llega un viajero que viene atraído por el turismo para conocer la ciudad de Sevilla, debido a que tuvo la oportunidad de conocer la promoción de la misma en su país por medio de la televisión y la prensa. El turista dialoga con el consejero solicitándole información sobre comidas, costumbres y visitas.
#1:2 Camino
Unidad 2  Preguntar el camino
#2:1.Mad
2.1  Madrid
	En el centro de Madrid se tropiezan  por casualidad dos amigos. Uno de ellos quiere ir al supermercado de ***, para hacer unas compras pero no sabe cómo llegar. El otro –que llega tarde a la oficina– le dice por dónde tiene que ir.
次が実行結果（段落で改行）です。

次は段落と文末で改行した結果です。

マクロプログラム
フォーム

	lblインポート
	【読込】クリップボードの内容をシートに読み込みます。

	opt段落で改行
	段落で区切ります。

	opt文末で改行
	文で区切ります。

	opt段落と文末で改行
	段落と文末で区切ります。

	cmd実行
	◎実行

	cmd削除
	▲削除

	cmd終了
	■終了

	lbl進行
	進行

	lblタイム
	タイム



Public Sub UserForm_Activate()
  Call ◎ユーザーフォーム実行
End Sub

Public Sub cmdExec_Click()
  Call ◎実行
End Sub

Private Sub cmdDelete_Click()
  Call ◎削除
End Sub

Private Sub cmdEnd_Click()
  Call ◎終了
End Sub

Private Sub UserForm_QueryClose(Cancel As Integer, CloseMode As Integer)
  Call ◎ユーザーフォームクローズ
End Sub
標準モジュールのコード
共通
Option Explicit

Public Declare Function FindWindow Lib "user32.dll" Alias "FindWindowA" _
  (ByVal lpClassName As String, ByVal lpWindowName As String) As Long 'フォームの取得
Public Declare Function GetWindowLong Lib "user32.dll" Alias "GetWindowLongA" _
  (ByVal handlewindow As Long, ByVal nIndex As Long) As Long 'ボタンを制御

Public Declare Function SetWindowLong Lib "user32.dll" Alias "SetWindowLongA" _
  (ByVal handlewindow As Long, ByVal nIndex As Long, ByVal dwNewLong As Long) As Long 'メニューバーの設定
Public Declare Function DrawMenuBar Lib "user32" (ByVal handlewindow As Long) As Long 'メニューバーの設定
        
Public Const GWL_STYLE = (-16) 'ウィンドウのスタイルを取得
Public Const WS_THICKFRAME = &H40000 'ウィンドウのサイズ変更
Public Const WS_MINIMIZEBOX = &H20000 '最小化ボタン

Public varStart, ShCnt% '開始時間、実行時のシート数
Public objRE As Object, Match As Object, Matches As Object '正規表現オブジェクト
Public i&, j& '作業用

Sub IMPORTE() 'ユーザーフォームを出現
  Dim handlewindow As Long 'フォームのハンドル
  Dim fButton As Long 'ボタンのスタイル
  Dim fReturn As Long '最小化状態からの復帰
  Dim fCaption As String 'フォームオブジェクトのキャプション名

  fCaption = frmI.Caption ' フォームのオブジェクト名からキャプション名を取得
  
  frmI.Show vbModeless 'フォームを表示
  
  handlewindow = FindWindow("ThunderDFrame", fCaption) 'フォームのハンドルを取得
  fButton = GetWindowLong(handlewindow, GWL_STYLE) 'ウィンドウに関する情報を取得
  fButton = fButton Or WS_MINIMIZEBOX '最小化ボタン
  fReturn = SetWindowLong(handlewindow, GWL_STYLE, fButton) 'フォームに指定したボタンを設定
  fReturn = DrawMenuBar(handlewindow) 'フォームのメニューバー外枠
End Sub

Sub ColumnsAutofit() '行幅・行高を自動調整
  ActiveSheet.UsedRange.Font.Name = frmI.cmbFontName.Value 'フォント名
  ActiveSheet.UsedRange.Font.Size = frmI.cmbFontSize.Value 'フォントサイズ
  ActiveSheet.UsedRange.Columns.ColumnWidth = Val(frmI.txtColMax) '列幅を最大値に
  ActiveSheet.UsedRange.Columns.AutoFit '列幅を自動調整
  ActiveSheet.UsedRange.Rows.AutoFit '行高を自動調整

  For i = 1 To Columns.Count
    If Columns(i).ColumnWidth > Val(frmI.txtColMax) Then '列幅が指定より大ならば
      Columns(i).ColumnWidth = Val(frmI.txtColMax) '列幅調整
    End If
  Next

  Selection.NumberFormatLocal = "@" '書式＝文字列
  Selection.HorizontalAlignment = xlLeft '左詰め
  Selection.WrapText = True '折り返し

  ActiveWindow.SplitRow = 1 'ウィンドウ枠を1行目で固定
  ActiveWindow.FreezePanes = True
End Sub

Sub ◎ユーザーフォーム実行()
  シート数% = Sheets.Count
  
  Set 正規表現o = CreateObject("VBScript.RegExp") '正規表現オブジェクトを生成
  正規表現o.Global = True: 正規表現o.IgnoreCase = True '文字列全体を検索、大小文字区別なし
  
  frmT.cboFontName.AddItem "Times New Roman": frmT.cboFontName.AddItem "Century"
  frmT.cboFontName.AddItem "Arial": frmT.cboFontName.AddItem "Courier New"
  frmT.cboFontName.AddItem "ＭＳ 明朝": frmT.cboFontName.AddItem "ＭＳ ゴシック"
  frmT.cboFontName.ListIndex = 0 '初期値
  
  For i = 1 To 9 '9まで繰り返す
    frmT.cboFontSize.AddItem i + 8 'フォントサイズ
  Next
  frmT.cboFontSize.ListIndex = 3
  
  For i = 1 To 26
    frmT.cboキー1.AddItem Chr$(64 + i) 'ソート列文字
    frmT.cboキー2.AddItem Chr$(64 + i) 'ソート列文字
    frmT.cboキー3.AddItem Chr$(64 + i) 'ソート列文字
  Next
  
  frmT.cboキー1.ListIndex = 0 '初期値
  
  frmT.cbo順1.AddItem "昇順": frmT.cbo順1.AddItem "降順": frmT.cbo順1.ListIndex = 0 '初期値
  frmT.cbo順2.AddItem "昇順": frmT.cbo順2.AddItem "降順": frmT.cbo順2.ListIndex = 0 '初期値
  frmT.cbo順3.AddItem "昇順": frmT.cbo順3.AddItem "降順": frmT.cbo順3.ListIndex = 0 '初期値
  
Sub ◎更新ボタン()
  frmT.lst入力シート.Clear '入力シート名ボックスをクリア

  For i = 1 To Sheets.Count 'シート数まで繰り返す
    frmT.lst入力シート.AddItem i & "_" & Sheets(i).Name '入力シート名を追加
  Next
End Sub

Sub ◎実行ボタン()
  On Error GoTo エラー
  Application.EnableCancelKey = xlErrorHandler 'Escキー押下時にエラーを送る
  
  開始時間v = Time
  
  Application.ScreenUpdating = False '画面表示を更新しない

  入力シート番号% = ActiveSheet.Index
  選択列番号% = Selection.Column
  
  入力行列v = ActiveSheet.UsedRange '範囲を取得
'    入力行列v = ActiveCell.CurrentRegion.SpecialCells(xlCellTypeVisible) 'フィルターの可視データを入力
  
  入力行数& = UBound(入力行列v, 1): 入力列数% = UBound(入力行列v, 2) '行数：列数
  
  出力行番号& = 0 '正規表現の数、出力シートの行数を初期化
  
  frmT.lblタイム = "実行中"
   
  Select Case frmT.MultiPage1.Value 'マルチページの番号でケースを選択
    Case 1: Call ■インポート
    Case 2: Call ■準備
    Case 3: Call ■シート離合
    Case 4: Call ■列離合
    Case 5: Call ■ソート
  End Select
  
  Application.ScreenUpdating = True '画面表示を更新
  
  frmT.lblタイム = Minute(Time - 開始時間v) & "分" & Second(Time - 開始時間v) & "秒" '経過時間
Exit Sub
エラー:
  MsgBox Error(Err) 'エラーメッセージ
End Sub

Sub ◎削除ボタン()
  If ActiveSheet.Index <= シート数% Then '起動時のシートならば…
    If MsgBox("起動したときのシート数以下のシートですが、削除しますか？", 1) = 1 Then 'OK
      シート数% = シート数% - 1 '起動時シート数をデクリメント
    Else
      Exit Sub
    End If
  End If
    
  Application.DisplayAlerts = False '確認しない
  ActiveSheet.Delete '現シートを削除
  Application.DisplayAlerts = True '確認する
End Sub

Sub ◎ユーザーフォーム終了()
  ◎終了ボタン
End Sub

Sub ◎終了ボタン()
  End '終了
End Sub

Sub 列自動調整()
  Dim 列%
  
  ActiveSheet.UsedRange.Font.Name = frmT.cboFontName.Value 'フォント名
  ActiveSheet.UsedRange.Font.Size = frmT.cboFontSize.Value 'フォントサイズ
  ActiveSheet.UsedRange.ColumnWidth = Val(frmT.txtColMax) '列幅調整
  ActiveSheet.UsedRange.Columns.AutoFit '列幅を自動調整
  
  For 列% = 1 To ActiveSheet.UsedRange.Columns.Count
    If Columns(列%).ColumnWidth > Val(frmT.txtColMax) Then
      Columns(列%).ColumnWidth = Val(frmT.txtColMax) '列幅調整
    End If
  Next
  
  ActiveSheet.UsedRange.Rows.AutoFit '行高を自動調整
  ActiveSheet.UsedRange.WrapText = True '折り返し
  
  [A2].Select: ActiveWindow.FreezePanes = True 'ウィンドー枠を固定
End Sub
インポート
[1] オプション
  '***** [1] データを準備 *****
  
  Dim データ$, 入力列v, 連番&, 見出し列番号%, 見出し列数%, 見出し行v, 出力行列v, 出力行番号&, 段落番号&, 文番号&
    
  With New DataObject 'クリップボードのデータ
    .GetFromClipboard: データ$ = .GetText
  End With
  
  データ$ = ReplaceRE(データ$, "\s*\n\s*", "\n") '改行の前後の空白を除去
  データ$ = ReplaceRE(データ$, "\n+", "\n") '改行の連続→1個の改行
  
  データ$ = ReplaceRE(データ$, "\n", "\n@#$") '段落頭をマーク
  データ$ = "@#$" & データ$ 'データ頭をマーク
  
  If Not frmT.opt段落で改行 Then
    データ$ = ReplaceRE(データ$, "([。．？！]+)", "$1\n") '文末で改行
    データ$ = ReplaceRE(データ$, "([!?]+)[ ]+(.*?)", "$1\n$2") '文末で改行
    データ$ = ReplaceRE(データ$, "([^\d]+?)(\.+)[ ]+([^\d]+?)", "$1$2\n$3") '文末で改行
  End If
  
  データ$ = ReplaceRE(データ$, "\n+", "\n") '改行の連続→1個の改行
はじめに、クリップボード内のデータをデータ$に代入します。
文末改行のオプションがあれば、単独の「。．？！」".?!"の後に改行コードを挿入します。
スペース、タブ、フォームフィードなどの空白文字を空白に置換し、改行前後の空白（連続）を除去し、改行の連続を1個の改行にします。
   '***** [2] 出力行列を作成 *****
  
  入力列v = Split(データ$, vbCr)
  
  For i = 0 To UBound(入力列v) - 1
    見出し列番号% = Val(ReplaceRE("A" & 入力列v(i), ".*#([1-9]+?):.*", "$1"))
    
    If 見出し列番号% > 見出し列数% Then 見出し列数% = 見出し列番号%
  Next i
  
  ReDim 見出し行v(1 To 見出し列数% + 2)
  ReDim 出力行列v(1 To UBound(入力列v) + 1, 1 To 見出し列数% + 4)
  
  出力行番号& = 1
  
  For i = 0 To UBound(入力列v) - 1
    見出し列番号% = Val(ReplaceRE("A" & 入力列v(i), ".*#([1-9]+?):.*", "$1"))
    
    If 見出し列番号% > 0 Then '見出しならば…
      見出し行v(見出し列番号%) = ReplaceRE(入力列v(i), "#[1-9]+:(.*)", "$1") '見出しを抽出
      見出し行v(見出し列番号%) = Replace(見出し行v(見出し列番号%), "@#$", "") 'マーク除去
      段落番号& = 0: 文番号& = 0 '初期化
    
      For j = 見出し列番号% + 1 To 見出し列数%
        見出し行v(j) = "-"
      Next
    
    Else 'テキストならば…
      出力行番号& = 出力行番号& + 1
      
      For j = 1 To 見出し列数%
        出力行列v(出力行番号&, 2 + j) = 見出し行v(j)
      Next
    
      If InStr(入力列v(i), "@#$") > 0 Then '段落頭ならば…
        段落番号& = 段落番号& + 1: 文番号& = 0
      End If
      
      連番& = 連番& + 1: 文番号& = 文番号& + 1
      出力行列v(出力行番号&, 1) = 連番&
      出力行列v(出力行番号&, 2) = Replace(入力列v(i), "@#$", "") 'テキスト：マーク除去
      出力行列v(出力行番号&, 見出し列数% + 3) = 段落番号&
      出力行列v(出力行番号&, 見出し列数% + 4) = 文番号&
    End If
    
    frmT.lbl進行 = 出力行番号&
  Next
　データ$を改行でスプリットし入力列vを用意します。
　はじめのFor .. Nextで属性見出しの最大列番を探します。このとき入力列の内容の前にAという文字を加えるのは、数字で始まるデータの誤読を防ぐためです。ReDimで、この最大列番を使って、出力行と出力面の規模を設定します。
　次のFor i… Nextで出力面を作成します。出力列番を探し、これがゼロ以上であれば属性見出しになるので、その列番に従って出力行を用意します。同時に文番を初期化し、該当する出力列番以降の出力行にピリオド( . )を補充します。出力列番がゼロであれば、テキストなので、For j … Nextで属性を出力面に代入し、文番をインクリメントして、文番と入力列を出力面に代入します。室力行番で進行を表示します。
 '***** [3] 出力 *****
  
  If 出力行番号& = 1 Then MsgBox "出力がありません。": Exit Sub
  
  出力行列v(1, 1) = "連番"
  出力行列v(1, 2) = "テキスト"
  
  For i = 1 To 見出し列数%
    出力行列v(1, 2 + i) = "見出し-" & i
  Next
  
  出力行列v(1, 見出し列数% + 3) = "段落"
  出力行列v(1, 見出し列数% + 4) = "文"
  
  Worksheets.Add After:=Sheets(Sheets.Count) 'シートを末尾に追加
  Range([A1], Cells(UBound(出力行列v, 1), UBound(出力行列v, 2))) = 出力行列v
  [A2].Select: ActiveWindow.FreezePanes = True 'ウィンドー枠を固定
  
  ActiveSheet.UsedRange.Font.Name = frmT.cboFontName.Value 'フォント名
  ActiveSheet.UsedRange.Font.Size = frmT.cboFontSize.Value 'フォントサイズ
  ActiveSheet.UsedRange.Columns.AutoFit '列幅を自動調整
  
  Columns(2).Select 'テキスト行選択
  
  If Selection.Columns.ColumnWidth > Val(frmT.txtColMax) Then '列幅が指定より大ならば
    Selection.Columns.ColumnWidth = Val(frmT.txtColMax) '列幅調整
  End If
  
  Selection.NumberFormatLocal = "@" '書式＝文字列
  Selection.HorizontalAlignment = xlLeft '左詰め
  Selection.WrapText = True '折り返し

  If frmT.opt段落で改行 Then Columns(見出し列数% + 4).Delete Shift:=xlToLeft
  If frmT.opt文末で改行 Then Columns(見出し列数% + 3).Delete Shift:=xlToLeft

  [A1].Select
End Sub
　出力行番がゼロのときは、出力がないことをメッセージボックスに表示して終了します。
  最大列番がゼロより大きいときは、文番タイトルを出力面に代入します。次にシートを追加して、出力面をシートのレンジに出力します。テキストのある行を選択して、列を自動調整します。

Private Function ReplaceRE$(ByVal 対象文字列$, 検索文字列$, 置換文字列$)
'＝関数：正規表現による置換
  検索文字列$ = Replace(検索文字列$, "\t", vbTab)
  検索文字列$ = Replace(検索文字列$, "\n", vbCr)
  正規表現o.Pattern = 検索文字列$
  置換文字列$ = Replace(置換文字列$, "\t", vbTab)
  置換文字列$ = Replace(置換文字列$, "\n", vbCr)
  ReplaceRE$ = 正規表現o.Replace(対象文字列$, 置換文字列$)
[bookmark: _GoBack]End Function
[bookmark: _Toc262039549][bookmark: _Toc264471443][bookmark: _Toc264516214][bookmark: _Toc264787839]準備
　「準備」のタブに移動してください。タブ「資料」の中で言語資料整理の準備をします。
[image: ]
準備
[bookmark: _Toc262039550][bookmark: _Toc264471444][bookmark: _Toc264516215][bookmark: _Toc264787840]　空白セルに文字を埋める
　Excelでソートするときに、空白セルがあると連続したデータとして扱われません。その空白セルに文字を入れると、連続してソートされます。その文字として1文字の空白（スペース）でもかまいません。
出力：
[image: ]
列を自動調整
「説明」タブで設定した出力フォント、サイズ、最大列幅に調整します。
出力：
[image: ]
[bookmark: _Toc262039552][bookmark: _Toc264471446][bookmark: _Toc264516217][bookmark: _Toc264787842]並列配置
　縦に並んだデータをそのA列とB列の内容にしたがって横に並べかえます。ちょうど2次元の座標のように、A列で行を決め、B列で列を決めます。これは日本語と英語などを比較しながら分析するために有用です。このような形式の言語資料をパラレルコーパスと呼びます（→コラム）。
入力：
[image: ]
出力：
[image: ]
[image: ]
直列配置
　逆に、横に並んだ資料を縦に並び替えるプログラムも有用です。縦に並べ替えることによって、各行のデータによって並べ替えができるようになります。
入力：並列配置の出力
出力：並列配置の入力

[bookmark: _Toc262039553][bookmark: _Toc264471447][bookmark: _Toc264516218][bookmark: _Toc264787843]マクロプログラム
フォーム

標準モジュール
Sub ■準備()
  If frmT.opt空白セルに文字付与 Then ●空白セルに文字付与
  If frmT.opt列自動調整 Then ●列自動調整
  If frmT.opt並列配置 Then ●並列配置
  If frmT.opt直列配置 Then ●直列配置

  Call 列自動調整
  frmT.lbl進行 = 入力行数& '進行
End Sub

Sub ●空白セルに文字付与()
  If frmT.chk新シート Then ActiveSheet.Copy After:=Sheets(Sheets.Count) '新シートにコピー
  
  出力行列v = ActiveSheet.Columns(選択列番号%) '選択列
  
  For i = 2 To 入力行数& '使用中の行数まで繰り返す
    If Trim(出力行列v(i, 1)) = "" Then 出力行列v(i, 1) = frmT.txt付与文字
  Next
  
  Range(Cells(1, 選択列番号%), Cells(入力行数&, 選択列番号%)) = 出力行列v '一括出力
End Sub
    
Sub ●列自動調整()
  If frmT.chk新シート Then ActiveSheet.Copy After:=Sheets(Sheets.Count) '新シートにコピー
End Sub

Sub ●並列配置()
  ReDim 出力行列v(1 To Application.Max([A:A]), 1 To Application.Max([B:B]))
  
  For i = 2 To 入力行数&
    出力行列v(入力行列v(i, 1), 入力行列v(i, 2)) = 入力行列v(i, 3)
  Next
    
  Sheets.Add After:=Sheets(Sheets.Count) '新シートを追加
  Range([A1], Cells(UBound(出力行列v, 1), UBound(出力行列v, 2))) = 出力行列v
End Sub
  
Sub ●直列配置()
  ReDim 出力行列v(1 To 入力行数& * 入力列数% + 1, 1 To 3)
  
  出力行列v(1, 1) = "行"
  出力行列v(1, 2) = "列"
  出力行列v(1, 3) = "データ"
  
  出力列番% = 1
  
  For i = 1 To 入力行数&
    For j = 1 To 入力列数%
      出力列番% = 出力列番% + 1
      出力行列v(出力列番%, 1) = i
      出力行列v(出力列番%, 2) = j
      出力行列v(出力列番%, 3) = 入力行列v(i, j)
    Next
  Next
    
  Sheets.Add After:=Sheets(Sheets.Count) '新シートを追加
  Range([A1], Cells(UBound(出力行列v, 1), UBound(出力行列v, 2))) = 出力行列v
End Sub
シートの結合と分離
[image: ]
複数のシートを結合
　複数のシートを１つのシートに連続して結合させます。リストボックスで複数のシートを選択してください。初期設定では「１行目はタイトル」にチェックをいれてあります。これによってそれぞれのシートで繰り返されるタイトル行が出力シートでは最初のシートの最初の１行だけとなります。タイトル行がなく、データが１行目からあるシートを結合するときは、このチェックを外してください。
出力:
[image: ]
選択列によってシートを分離
　選択された列にあるデータを基準にして異なるシートに分離します。選択されたデータの種類が多いと、その数だけシートが作成されてしまうので、最大シート数の初期値を10としてあります。
[bookmark: _Toc262039554][bookmark: _Toc264471449][bookmark: _Toc264516220][bookmark: _Toc264787845]入力：3.1の出力
出力：3.1の入力
[bookmark: _Toc262039560][bookmark: _Toc264471455][bookmark: _Toc264516226][bookmark: _Toc264787851]マクロプログラム
フォーム
[image: ]

標準モジュール
Sub ■シート離合()
  If frmT.optシート結合 Then ●シート結合
  If frmT.optシート分離 Then ●シート分離
End Sub

Sub ●シート結合()
  Dim タイトル行%

  For i = 0 To frmT.lst入力シート.ListCount - 1 'リストボックスの項目数まで繰り返す
    If frmT.lst入力シート.Selected(i) Then 'リストボックスの項目が選択されていれば…
      Sheets(i + 1).Activate '入力シートをアクティべート
      入力行数& = ActiveSheet.UsedRange.Rows.Count '現シートの行数
      入力列数% = ActiveSheet.UsedRange.Columns.Count '現シートの列数
      タイトル行% = IIf(frmT.chkタイトル行, 2, 1) '最初の行がタイトルならばj=2, そうでなければj=1
      入力行列v = ActiveSheet.UsedRange '現シート使用範囲
              
      If 出力行番号& = 0 Then '最初のシートならば
        Sheets(i + 1).Copy After:=Sheets(Sheets.Count) '新シートにコピー
        Columns(1).Insert '列を挿入
        Cells(1, 1) = "Sheet" 'タイトル
        Range(Cells(2, 1), Cells(入力行数&, 1)) = Sheets(i + 1).Name 'シート名
        'シート名を一括出力
        出力行番号& = ActiveSheet.UsedRange.Rows.Count '出力シートの行数
      
      Else '2枚目以降のシートならば
        Sheets(Sheets.Count).Activate '新シートをアクティべート
      
        For j = タイトル行% To 入力行数&
          Cells(出力行番号& + j - タイトル行% + 1, 1) = Sheets(i + 1).Name 'シート名
          
          For k = 1 To 入力列数%
            Cells(出力行番号& + j - タイトル行% + 1, k + 1) = 入力行列v(j, k) 'データ
          Next
        Next
      End If
      
      出力行番号& = ActiveSheet.UsedRange.Rows.Count '出力シートの行数
    End If
    
    frmT.lbl進行 = i + 1 & "/" & frmT.lst入力シート.ListCount '進行
  Next
  
  Call 列自動調整
End Sub

Sub ●シート分離()
  Dim 追加シート数%

  ActiveSheet.Copy After:=Sheets(Sheets.Count) 'シートをコピー
  ActiveSheet.UsedRange.Sort Header:=xlYes, MatchCase:=True, Key1:=Columns(選択列番号%) 'ソート
  入力シート番号% = ActiveSheet.Index
  
  For i = 2 To 入力行数& '行数まで繰り返す
    If 入力行列v(i, 選択列番号%) <> 入力行列v(i - 1, 選択列番号%) Then
    '前のデータと異なるとき…
      追加シート数% = 追加シート数% + 1
      
      If 追加シート数% > Val(frmT.txt最大シート数) Then
        MsgBox "最大追加シート数を超えました。"
        
        For j = 1 To 追加シート数%
          Application.DisplayAlerts = False '確認しない
          Sheets(Sheets.Count).Delete '作業用シートを削除
          Application.DisplayAlerts = True '確認
        Next
        Exit Sub
      End If
      
      Call 列自動調整
      Sheets.Add After:=Sheets(Sheets.Count)  '新シートを追加
      Sheets(入力シート番号%).Rows(1).Copy Rows(1) 'タイトルをコピー
      出力行番号& = 1 '出力行数を初期化
    End If
      
    出力行番号& = 出力行番号& + 1 '出力行をインクリメント
    Sheets(入力シート番号%).Rows(i).Copy Rows(出力行番号&) 'データをコピー
    frmT.lbl進行 = i & "/" & 入力行数& '進行
  Next
    
  Call 列自動調整
  Application.DisplayAlerts = False '確認しない
  Sheets(入力シート番号%).Delete '作業用シートを削除
  Application.DisplayAlerts = True '確認
End Sub
列の離合
複数の列のデータを特定のマークを間に介して結合します。また、逆に１列の中にある文字列を、特定のマークによって複数の列や行に分離します[footnoteRef:5]。 [5:  このマークは、一般に区切り記号（セパレーター）と呼ばれます。] 

[image: ]
列離合
　別々の列に分かれたデータをまとめたり、１つの列に入ったデータを複数の列に分けたりことが必要になることがあります。たとえば異なる列に配置された動詞の変化形と代表形をアンダーバーでつないで別の列に出力させておくと、後で分析がしやすくなります。
[bookmark: _Toc262039561][bookmark: _Toc264471456][bookmark: _Toc264516227][bookmark: _Toc264787852]列の結合
 隣接した複数の列を選択し、それらをマークをつけて結合します。初期設定では出力で結合する対象列を削除します。
入力：AB列を選択
[image: ]
出力：
[image: ]
[bookmark: _Toc262039562][bookmark: _Toc264471457][bookmark: _Toc264516228][bookmark: _Toc264787853]横に分離
　１列の中にあるデータをマークによって要素の数だけ複数の列に分離します。
入力：2.3.1の出力
出力：2.3.1の入力
[bookmark: _Toc262039563][bookmark: _Toc264471458][bookmark: _Toc264516229][bookmark: _Toc264787854]縦に分離
　マークによって１列を複数の列に分離し、要素の数だけ縦に並べて出力します。次は分離するマークとしてコンマ(,)を指定しました。
入力：2.3.1の出力シート
出力：
[image: ]
マクロプログラム
フォーム

標準モジュール
Sub ■列離合()
  If frmT.chk新シート Then ActiveSheet.Copy After:=Sheets(Sheets.Count)
  '新シートに出力するときコピー
      
  If frmT.opt列結合 Then ●列結合
  If frmT.opt列横分離 Then ●列横分離
  If frmT.opt列縦分離 Then ●列縦分離
  
  Call 列自動調整
End Sub

Sub ●列結合()
  ReDim 出力行列v(1 To 入力行数&, 1 To 1)
    
  入力行列v = Selection '選択範囲
      
  For i = 1 To 入力行数& '行の終わりまで…
    出力行列v(i, 1) = 入力行列v(i, 1)
    
    For j = 2 To UBound(入力行列v, 2) '列数まで…
      出力行列v(i, 1) = 出力行列v(i, 1) & frmT.txtマーク & 入力行列v(i, j) '文字列を作成
    Next
  
    frmT.lbl進行 = i & "/" & 入力行数& '進行
  Next
    
  Selection.Delete Shift:=xlToLeft '選択行を削除
  
  Columns(選択列番号%).Insert '選択列の前に1列を挿入
  
  If ActiveSheet.UsedRange.Columns.Count = 1 Then 選択列番号% = 1
  Columns(選択列番号%).HorizontalAlignment = xlLeft '左詰め
  Columns(選択列番号%).NumberFormatLocal = "@"
  Range(Cells(1, 選択列番号%), Cells(入力行数&, 選択列番号%)) = 出力行列v '出力シートに一括出力
End Sub

Sub ●列横分離()
  配列$ = Split(Trim(入力行列v(2, 選択列番号%)), frmT.txtマーク) 'スプリット
  
  ReDim 出力行列v(1 To 入力行数&, 1 To UBound(配列$) + 1)
  
  For i = 1 To 入力行数& '使用中の行数まで繰り返す
    配列$ = Split(入力行列v(i, 選択列番号%), frmT.txtマーク) 'スプリット
    
    For j = 0 To UBound(配列$) '配列数まで繰り返す
      If j > 0 And frmT.optマーク先頭 Then 配列$(j) = frmT.txtマーク & Trim(配列$(j))
      If j < UBound(配列$) And frmT.optマーク末尾 Then 配列$(j) = Trim(配列$(j)) & frmT.txtマーク
      
      出力行列v(i, j + 1) = Trim(配列$(j)) '出力
    Next
    
    frmT.lbl進行 = i & "/" & 入力行数& '進行
  Next
  
  Columns(選択列番号%).Delete '元列を削除
  
  Range(Columns(選択列番号%), Columns(選択列番号% + UBound(配列$))).Insert
  '選択列の前に複数列を挿入
     
  Range(Cells(1, 選択列番号%), Cells(入力行数&, 選択列番号% + UBound(配列$))).NumberFormatLocal = "@"
 
  Range(Cells(1, 選択列番号%), Cells(入力行数&, 選択列番号% + UBound(配列$))) = 出力行列v '一括出力
End Sub
  
Sub ●列縦分離()
  配列$ = Split(Trim(入力行列v(2, 選択列番号%)), frmT.txtマーク) 'スプリット
  
  ReDim 出力行列v(1 To 入力行数& * (UBound(配列$) + 1), 1 To 入力列数%)
   
  For i = 1 To 入力列数%
    出力行列v(1, i) = 入力行列v(1, i) 'タイトル行を代入
  Next
  
  出力行番号& = 1
   
  For i = 2 To 入力行数&
    配列$ = Split(入力行列v(i, 選択列番号%), frmT.txtマーク) 'スプリット
    
    For j = 0 To UBound(配列$)
      出力行番号& = 出力行番号& + 1
      
      For k = 1 To 入力列数%
        出力行列v(出力行番号&, k) = 入力行列v(i, k)  '当該行を出力
      Next
      
      If j > 0 And frmT.optマーク先頭 Then 配列$(j) = frmT.txtマーク & Trim(配列$(j))
      If j < UBound(配列$) And frmT.optマーク末尾 Then 配列$(j) = Trim(配列$(j)) & frmT.txtマーク
      
      出力行列v(出力行番号&, 選択列番号%) = Trim(配列$(j)) '当該セルを出力
    Next
    
    frmT.lbl進行 = i & "/" & 入力行数& '進行
  Next

  Range([A1], Cells(UBound(出力行列v, 1), UBound(出力行列v, 2))).NumberFormatLocal = "@"
  
  Range([A1], Cells(UBound(出力行列v, 1), UBound(出力行列v, 2))) = 出力行列v '一括出力
End Sub

	要素・単位
　エクセルでは基本的に１つのセルに１つの要素を配置します。そのとき何を要素とするか、その単位を明らかにしておかなければなりません。そうでないと列の中に異質な要素が混じってしまい、後の分析が困難になるからです。
　動詞の変化形と代表形のように、異なる要素を１つにまとめて結合要素とすることもあります。これは、第Ⅱ部で扱う「分析」の対象がいつも１列に限られ、分析の単位がいつも１つになるからです。
　言語資料を分析するとき、合成語や成句などを１つの単位とするか、または複数の単位の結合とするかで、分析の結果が異なります。記述言語学(Descriptive Linguistics)の方法では、この「単位」ということについて非常に熱心に論じられ、その理論的な定義について論争が繰り返されました。要素の単位が決まらないと言語の記述ができないからです。これは理論が優先される演繹的方法です。
　一方、コンピュータで言語を処理するときには、多くの場合、書かれたテキストをそのまま使い、要素がつながるときの傾向を数量的に計算することで、単位としての結合要素を見いだしていく、という帰納的な方法が採られています。こちらのほうが実際的です。両者の立場の違いは学習・研究用コンピュータの出現以前・以後の時代差が反映されていると思います。



ソート
　データのソート（並べ替え）はエクセルで頻繁に行う作業ですが、ここで扱っている「統合ID」（コロンで区切られた数字列）はExcclによる並べ替えではできません。「ソート」のタブで、それを可能にしました。また、言語研究でよく使う逆引き配列による並べ替えもあると便利です。初期値は昇順ですが、「降順」を選択することができます。
[image: ]
[bookmark: _Toc262039565][bookmark: _Toc264471460][bookmark: _Toc264516231][bookmark: _Toc264787856]一般的な整列
　はじめに一般的な並べ替えによる整列をしてみましょう。昇順の場合は「第1キー」だけを選択します[footnoteRef:6]。 [6:  数字は文字の順番ではなく、数の大小を基準にします。たとえば、 2, 1, 11の昇順は1, 2, 11であり、1, 11, 2という（文字の）順番になりません。] 


出力：
[image: ]
[bookmark: _Toc262039566][bookmark: _Toc264471461][bookmark: _Toc264516232][bookmark: _Toc264787857]統合IDの整列
　統合IDによる並べ替えをします。統合IDの各要素は5桁までの大きさまで並べ替えられます。ここではシートGenを使い、対象となるA列を選択します。
出力：
[image: ]
[bookmark: _Toc262039567][bookmark: _Toc264471462][bookmark: _Toc264516233][bookmark: _Toc264787858]逆引き配列による整列
対象の文字列の逆引き配列による整列をします。語尾による語形変化などの分析に役立ちます。
出力：
[image: ] (...)

標準モジュール
Sub ■ソート()
  Dim キー列番%(3), bol統合(3) As Boolean, bol逆引(3) As Boolean

  If frmT.chk新シート Then ActiveSheet.Copy After:=Sheets(Sheets.Count) '新シートにコピー
  
  bol統合(1) = frmT.opt統合1: bol統合(2) = frmT.opt統合2: bol統合(3) = frmT.opt統合3 '統合
  bol逆引(1) = frmT.opt逆引1: bol逆引(2) = frmT.opt逆引2: bol逆引(3) = frmT.opt逆引3 '逆引
  
  キー列番%(1) = frmT.cboキー1.ListIndex + 1 '対象列1
  キー列番%(2) = frmT.cboキー2.ListIndex + 1 '対象列2
  キー列番%(3) = frmT.cboキー3.ListIndex + 1 '対象列3
    
  ReDim 出力行列v(1 To 入力行数&, 1 To 3) As Variant '作業行用配列のメモリー
  
  For i = 1 To 3 'キーの数だけ繰り返す
    If キー列番%(i) > 0 Then 'キーが選択されていれば…
      If bol統合(i) Then '統合IDの場合…
        For j = 2 To 入力行数& '使用中の行数まで繰り返す
          文字列$ = 入力行列v(j, キー列番%(i))
          配列$ = Split(文字列$, ":") '統合IDを":"でスプリット
          文字列$ = "[" '文字として扱う
      
          For k = 0 To UBound(配列$) '配列数まで繰り返す
            文字列$ = 文字列$ & Right$("0000" & Trim(配列$(k)), 5) '文字列を作成
          Next
        
          出力行列v(j, i) = 文字列$ '文字列を出力
        Next
      
        Columns(キー列番%(i)).HorizontalAlignment = xlLeft '左詰め
      
      ElseIf bol逆引(i) Then '逆引きの場合…
        For j = 2 To 入力行数& '使用中の行数まで繰り返す
          出力行列v(j, i) = StrReverse(Trim(入力行列v(j, キー列番%(i)))) 'トリム、逆引き
        Next
        
        Columns(キー列番%(i)).HorizontalAlignment = xlRight '右詰め
      
      Else 'それ以外ならば
        For j = 2 To 入力行数& '使用中の行数まで繰り返す
          出力行列v(j, i) = 入力行列v(j, キー列番%(i)) '対象列を移動
        Next
      End If
    Else
      For j = 2 To 入力行数& '使用中の行数まで繰り返す
        出力行列v(j, i) = 0 'ダミー
      Next
    End If
  
    frmT.lbl進行 = i & "/" & 3 '進行
  Next i
  
  Range(Cells(1, 入力列数% + 1), Cells(入力行数&, 入力列数% + 3)) = 出力行列v '一括出力
  
  ActiveSheet.UsedRange.Sort Header:=IIf(frmT.chkソートタイトル行, xlYes, xlNo), _
    MatchCase:=True, _
    Key1:=Columns(入力列数% + 1), _
    Order1:=IIf(frmT.cbo順1.ListIndex = 1, xlDescending, xlAscending), _
    Key2:=Columns(入力列数% + 2), _
    Order2:=IIf(frmT.cbo順2.ListIndex = 1, xlDescending, xlAscending), _
    Key3:=Columns(入力列数% + 3), _
    Order3:=IIf(frmT.cbo順3.ListIndex = 1, xlDescending, xlAscending) 'ソート
  
  Range(Columns(入力列数% + 1), Columns(入力列数% + 3)).Delete 'ソート用列を削除

  Call 列自動調整
End Sub

	並べ替え・逆引き
　Excelの並べ替えの機能は強力で、かなり大きなデータでも短時間で出力します。TABLASでもコードの中でその機能を利用しています。
　現在のようにコンピュータが手軽に使えなかった時代には、言語資料はカードに書き写していました。エクセルの１行にあたる部分が１枚のカードになります。カードの束をいろいろな基準で並べ替えたり、広い空間に配置したりしてカードごとの関係を調べました。カードの数が少なければそれほど時間もかからないのですが、言語資料は膨大な量になることがよくあります。
　そこでエクセルの登場です。しかし、セルにデータを１つずつ書き込んでいく作業はカードを作るのとほとんど変わりがありません。できるならば、一連のテキストデータから自動的な作業をして、分析用のシートにしておきたいものです。
　分析用シートができてしまえば、それからの作業はアイデア次第でさまざまな出力が得られます。並べ替えもその一つです。並べ替えの出力は単純なので、言語資料の分析にはあまり役に立たないと思われるかもしれません。でも、データが昇順・降順に並んでいる、ということには次のような大きな利点があります。
　・検索がしやすい
　・同じデータが集まっている
　・類似したデータが近くにある
　・同一または類似したデータの量が視覚的にとらえられる
　・不規則・不良なデータが判明する
　そして、キーを増やせば下位分類ができます。第1キーが大分類、第2キーが大分類中の中分類、第3キーが中分類中の小分類になります。
　文字列を昇順に並べたときも、これと同じ原則になっていることがわかります。たとえば次のリストでは、
[image: ]
　aという文字の大分類の次に、ゼロ、b, c, dという2文字目が基準となる中分類があり、さらにdという中分類には3文字目のa, dという小分類があります。さらにaという小分類の中ではe (=Adae)とm(=Adam)に分かれています。私たちが辞書を使うときは基本的にこのような順で検索をしています。
　次は逆引き・昇順の出力です。
[image: ]
　ここでの第1キーは語尾のaとなります。第2キーは語尾から2番目の文字でゼロ、d, e, iが並んでいます。第3キーは語尾から3番目の文字です。第2キーまでが同じiaの中でdiaとliaが並んでいます。volatiliaとreptiliaを比較するとtiliaが同じなので第6キー(語尾から数えて6番目の文字)のaとpが比較されvolatiliaが昇順では先になります[footnoteRef:7]。このように逆引きではキーを語尾から数えていくので、右揃えのほうがわかりやすいのです。 [7:  実際のマクロによるプログラミングでは正引きでも逆引きでも文字列全体を比較し、ここでしたような個別の文字の比較をしません。] 

　この逆引き配列は語尾について、先に挙げた5つの利点があてはまりますから、たとえば活用形、接尾辞、英語の’sなどの付加などを調べるときに便利です。

